
Serial Communication

For two devices to communicate, regardless of whether they are computers (e.g. your 
Mac/PC or Raspberry PI), microcontrollers (e.g. Arduino), or any other form of 
integrated circuit, the devices need a method of communication (e.g. voltage on wires 
or flashes of light) and an agreed-upon language (a.k.a. protocol).

The most common form of communication between electronic devices is serial 
communication. Communicating serially involves sending a series of digital pulses 
back and forth between devices at a mutually agreed-upon rate.

In asynchronous serial communication the sender transmits pulses representing the 
data at the agreed–upon data rate, and the receiver listens for pulses at that same 
rate.

Note: Unlike SPI, which we saw last lecture, in asynchronous serial communication there 
isn’t a common clock; instead, both devices have their own clock and agree on a rate to 
which to set their clocks.



Serial Communication

To set up a basic serial communication there need to be three wire connections 
between the two devices:

• a common ground allows the devices to measure voltage against same reference (0V);

• one wire for the sender to transmit data to the receiver (TX for the sender);

• one wire for the receiver to transmit date to the sender (RX for the sender).

If the speed is set at 9600 bits per second (a.k.a. 9600 baud), then the receiver will 
continually read the voltage and every 1/9600th of a second. The receiver will interpret 
the voltage as a new bit of data, for example:

If the voltage is high (+5V), it will interpret that bit of data as a 1.

If it is low (0V), it will interpret that bit of data as a 0.

Note: some serial communication systems invert the meaning of this voltage so that 
high becomes 0 and low becomes 1.



Serial Communication

At 9600 baud, a serial connected can send 1.2Kb per second, or 72Kb per minute. This 
is not a lot of data, i.e., if you wanted to send an uncompressed greyscale image over 
this connection every second you would have to limit its size to 34x34 pixels.

9600 baud is a commonly used speed when starting a project with relatively little data 
to send, like the examples we’re going to look at today. But there are a number of other 
common speeds that can be used to transmit data over a serial connection like this. 

Arduino hardware can handle up to 115,200 bits per second, allowing it to send and receive 
up to ~14Kb per second. This would increase the size of the uncompressed greyscale image 
we could send over this connection to 120x120 pixels.

Slower speeds are good for sending small amounts of data. They have the advantage 
that (1) the signals tend to be more reliably transmitted over the wire, i.e., the pulses 
are resilient to interference because they are relatively long, and (2) the sender and 
receiver don’t have to be very fast, or have large data buffers, to communicate at a 
slow speed.



Serial Communication

Here’s an example of one byte of data, representing the number 90, is exchanged. 
First, the number from the decimal representation 90 to a binary representation. in 
binary, 90 is 01011010. So the sending device will pulse its transmit line as follows:

Both devices have to agree on the order of the bits. Usually the sender sends the 
highest bit (or most significant bit) first, and the lowest (or least significant bit) last.



Dimmer (Arduino Code)

const int ledPin = 9;

void setup() {
  Serial.begin(9600); // initialize the serial communication
  pinMode(ledPin, OUTPUT);
}

void loop() {
  byte brightness;
  if (Serial.available()) { // check if data has been sent
    brightness = Serial.read();
    analogWrite(ledPin, brightness);
  }
}



Dimmer (Processing Code)

import processing.serial.*;
Serial port;
 
void setup() {
  size(256, 150);
  println("Available serial ports:");
  println(Serial.list());
  port = new Serial(this, Serial.list()[0], 9600); // open serial port
}

void draw() {
  for (int i = 0; i < 256; i++) {
    stroke(i);
    line(i, 0, i, 150);
  }
  port.write(mouseX); // send data to the serial port
}



Graph (Arduino Code)

void setup() {
  Serial.begin(9600); // initialize the serial communication
}

void loop() {
  Serial.println(analogRead(A0));
  delay(2);
}



Graph (Processing Code)

import processing.serial.*;
Serial myPort;
int xPos = 1;

void setup() {
  size(400, 300);
  println(Serial.list());
  myPort = new Serial(this, Serial.list()[0], 9600);
  myPort.bufferUntil('\n');
  background(0);
}

void draw () { } // Everything happens in serialEvent()



Graph (Processing Code) cont.

void serialEvent(Serial myPort) {
  String inString = myPort.readStringUntil('\n');  // get the ASCII string
 
  if (inString != null) {
    inString = trim(inString); // trim off any whitespace
    float inByte = float(inString); // convert to an int
    inByte = map(inByte, 0, 1023, 0, height); // map to the screen height

    stroke(127,34,255);
    line(xPos, height, xPos, height - inByte); // draw the line

    if (xPos >= width) {
      xPos = 0;
      background(0); 
    } else {
      xPos++;
    }
  }
}



Exercise

Experiment with the serial communication examples given in the Arduino examples, 
like Dimmer and Graph, to see how Arduino and Processing can send and receive 
simple values.



Firmata

An alternative to writing our application–specific serial communication systems, like 
the Dimmer and Graph examples, is to use a general–purpose “firmware” on the 
Arduino, which provides access to the features of the Arduino, e.g., input/output pins, 
through a standard protocol.

Firmata is one such example of this type of general–purpose firmware designed for the 
Arduino family. The Arduino IDE comes with a few different versions of Firmata, that 
provide slightly different ranges of features, e.g., control of servos etc.

We can use the StandardFirmata implementation. Processing has an Arduino library that is 
designed to talk to Arduino hardware running the StandardFirmata firmware, and comes 
with a number of examples.



arduino_pwm (Processing Code)

import processing.serial.*;
import cc.arduino.*;

Arduino arduino;

void setup() {
  size(512, 200);
  arduino = new Arduino(this, Arduino.list()[0], 57600);
}

void draw() {
  background(constrain(mouseX / 2, 0, 255));
  arduino.analogWrite(9, constrain(mouseX / 2, 0, 255));
  arduino.analogWrite(11, constrain(255 - mouseX / 2, 0, 255));
}


